TIMEX sinclair

USERS GROUP OF CINCINNATI

Vol. 2 No. 7 August, 1984

NEXT MEETING:

August 26, 1984

September 23, 1984

All meetings are held at 2:00 P.M. in room 506, Crosley Tower, University of Cincinnati.

Things have been quite this last month, everyone must be on vacation. Send us a postcard!

+++ Timex has finally shipped the 2068 technical manual. I have mine.

+++ Jack Roberts still has 2020 printers and 2050 modems.

Picnic → See reverse side for map to Winton Woods and picnic area.

The First Annual T/S Users Group Picnic is Sunday August 26, 1984, 12 noon till 5:00+ Bring your own food and drinks and one item for the group to share. We will have a grill and charcoal. This will take the place of our August meeting. We will have it rain or shine.

We will try to get one of these areas:
1) Kestrel Point Shelter
2) Mallard Pond View Shelter
3) Island View

Please call Rick Johnson at 825-1449 or Bill Sieber at 353-3482 till 11:30 AM to confirm the location.
Everyone is invited, bring your family and recreational equipment. Sorry, no computers.

NOTE: A motor vehicle permit is required for Winton Woods. These may be obtained at Activity centers, Ranger stations, or the park entrance booth. $1.00/day or $3.00/yr.

****
****

Please submit all articles to:
Bill Bernard
6316 Firestone Dr.
Fairfield, OH 45014

CONTENTS

First Annual T/S Users Group Picnic Sunday 8/26/84 12:00 till 5:00± Bring your own food and drinks and one item for the group to share. We will have a grill and charcoal This will take the place of our August meeting. We will have it rain or shine.



We will try to get one of these areas in this order:

(3) → (2) → (1) →

Please call Rick Johnson at 825-1449 or Bill Sieber 353-3482 till 11:30 to confirm location on Sunday 8/26/84

Everyone is invited - Bring your family (No Computers) and recreational equipment.

Note Motor Vehicle Permit Required obtain at Activity Centers, Ranger Station,

OFFICERS:

Rick Johnson, President	825-1449
Kurt Albrecht, Treasurer	542-3921
Gary Szekeres, Secretary	331-8966

THIS SPACE WAS HELD UNTIL PRESS TIME FOR YOUR ARTICLE, BUT YOU PUT OFF SUBMITTING IT. WE CAN'T PUBLISH WHAT YOU DON'T SEND US.

WE ESPECIALLY NEED ARTICLES FOR THE ZX81 & T/S 1000/1500 COMPUTERS.

Cartoon is reprinted from the SINCUS NEWS a publication of the Sinclair Users Society of Johnson City, New York

Thanks to Paul Hill, SINCUS

T/S 2068 Tape Copy Utility

This machine language program can be used to copy programs, arrays, screens, and code from one tape to another. The program copies the entire program, array, etc into memory; and then writes it to the output tape. The program is designed to use two parallel control ports to start/stop both the input and output cassette tape recorders. However, the user may manually control the cassette tape recorders. The user could even use a single cassette tape recorder for both the input and output tapes.

To use this program, type:

1) LOAD "tapecopy"CODE

2) PRINT USER 23760

Note: The program is tucked into system RAM above the system variables, but below the stack.

The program begins by displaying a heading line on the TV and printer(if attached and on).

If the user has a parallel control port at address 127, bit 0 on should start the input cassette tape recorder; bit 1 on should start the output cassette tape recorder. The bits must be latched on until port 127 is again addressed. The program will automatically control the starting/stopping of both cassette tape recorders.

In a manual mode of operations, start the input cassette tape recorder when the heading line appears on the TV screen. The program will read the input tape and display the type(program, array, code), name, starting address/line number, and program size. When the copy? prompt appears on the lower screen, stop the input cassette tape recorder and start the output cassette tape recorder. Then reply 'Y' (upper or lower case is acceptable) to the prompt. The program will then write the program, array, etc to the output tape. When it is finished it will display an '*' under the column marked C(opied) on the TV screen; it will then write the line to the printer. Stop the output cassette tape recorder and start the input cassette tape recorder.

Repeat until finished.

If you reply anything but 'Y' to the copy? prompt the program will continue to read the input tape. Therefore the user must be sure the input cassette tape recorder continues to play and the output cassette tape recorder is stopped.

The user could use only one cassette tape recorder by switching input/output tapes and the play/record switches at the appropiate times.

The next two pages contain the Zeus assembler source for the tape copy program.

The back page contains a HEX dump of the program which could be POKEd into memory if you don't have an assembler.

```
00100         ORG 23760
00200         ENT
00300         CALL STOPT
00400         LD  HL,#803D
00500         PUSH HL
00600         CALL #6200 ;CLS
00700         CALL PRNTH
00800         CALL PAPRH
00900 AGAIN   CALL STRT1
01000         CALL EXROM
01100 READH   SCF
01200         LD  IX,HTYPE
01300         LD  A,0 ;READ LABEL
01400         LD  DE,17
01500         CALL 252
01600         JR  NC,READH
01700         CALL HOME
01800         LD  A,(HTYPE)
01900         SLA A ;TIMES 8 FOR
02000         SLA A ;INDEX INTO
02100         SLA A ;PTYPE TABLE
02200         LD  C,A
02300         LD  B,0
02400         LD  HL,PTYPE ;BASE
02500         ADC HL,BC ;+ INDEX
02600         LD  B,8 ;LENGTH
02700         CALL TV
02800         LD  HL,(DFCC) ;SCRN
02900         LD (SAVEL),HL ;ADDR
03000         LD  HL,HNAME
03100         LD  B,10 ;LENGTH
03200         CALL TV
03300         LD  HL,BLANK
03400         LD  B,1
03500         CALL TV
03600         CALL EXROM
03700         SCF
03800         LD  IX,(PROG) ;ADDR
03900         LD  A,255
04000         LD  DE,(HLEN) ;LEN
04100         CALL 252 ;READ TAPE
04200         PUSH AF
04300         CALL HOME
04400         CALL STOPT
04500         POP  AF
04600         JP NC,SKIP ;OK?-NO
04700         LD  A,(HTYPE)
04800         CP  0 ;BASIC PGM
04900         JR  NZ,CADDR ;NO
05000         LD  A,(HADDR+1)
05100         BIT 7,A ;STRT ADDR?
05200         JR  Z,CADDR ;YES
05300         LD  HL,BLANK
05400         LD  B,6 ;LENGTH
05500         CALL TV
05600         JR  PSIZE
05700 CADDR   LD  HL,(HADDR)
05800         CALL NUMER
05900         LD  HL,BLANK ;ONE
06000         LD  B,1 ;BLANK BYTE
06100         CALL TV
06200 PSIZE   LD  HL,(HLEN) ;PGM
06300         CALL NUMER ;SIZE
06400         LD  HL,KCOPY ;COPY?
06500         LD  B,6
06600         CALL PRNT1 ;LOWER
06700         LD  A,143
06800         LD HL,#8034 ;FLASHA
06900         PUSH HL
07000         CALL #6200
07100 GETIN   LD  HL,#801A
07200         PUSH HL
07300         LD  DE,300 ;CYCLES
07400         LD  HL,200 ;PERIOD
07500         CALL #6200 ;PARP
07600         HALT ;KEYBOARD WAIT
07700         LD  A,(FLAGS)
07800         BIT 5,A ;ANYTHING?
07900         JR  Z,GETIN ;NO
08000         RES 5,A ;FLG OFF
08100         LD  (FLAGS),A
08200         LD A,(LASTK) ;REPLY
08300         PUSH AF
08400         LD  HL,#8021
08500         PUSH HL
08600         CALL #6200 ;CLS BOT
08700         POP AF
08800         CP  "Y ;YES?
08900         JR  Z,COPY
09000         CP  "y ;YES?
09100         JP  Z,COPY
09200 SKIP    CALL ENDTV
09300         JP  AGAIN
09400 COPY    CALL STRT2 ;STRT TP
09500         CALL WAIT
09600         CALL WAIT
09700         CALL EXROM
09800         SCF
09900         LD  IX,HTYPE
10000         LD  A,0 ;WRITE LABL
10100         LD  DE,17 ;LENGTH
10200         CALL 104 ;WRITE TP
10300         JP NC,ERROR ;OK?-NO
10400         CALL HOME
10500         CALL WAIT
10600         CALL EXROM
10700         SCF
10800         LD  IX,(PROG) ;OUT
10900         LD  A,255
11000         LD  DE,(HLEN) ;LEN
11100         CALL 104 ;WRITE DAT
11200         JP NC,ERROR ;OK?-NO
11300         CALL HOME
11400         CALL STOPT
11500         LD  HL,ASTER
11600         LD  B,2
11700         CALL TV
11800         CALL ENDTV
11900         LD  B,8 ;# SCAN LNS
12000         LD  HL,(SAVEL)
12100         LD  A,L
12200         AND #E0 ;MODULO 32
12300         LD  L,A
12400         DI
12500 NEXTR   PUSH HL
12600         PUSH BC
12700         LD  DE,#8024
12800         PUSH DE
12900         CALL #6200 ;PRSCAN
13000         POP BC
13100         POP HL
13200         INC H ;NEXT LINE
13300         DJNZ NEXTR
13400         EI
13500         JP  AGAIN ;REPEAT
13600 EXROM   DI
13700         IN  A,(255)
13800         SET 7,A
13900         OUT (255),A ;EXROM
14000         IN  A,(244)
14100         LD  (HSAVE),A
14200         LD  A,1 ;CHUNK 0
14300         OUT (244),A
14400         EI
14500         RET
14600 HOME    DI
14700         LD  A,(HSAVE)
14800         OUT (244),A
14900         IN  A,(255)
15000         RES 7,A
15100         OUT (255),A
15200         EI
15300         RET
15400 PRNTH   CALL SLCT2 ;SCRN
15500         LD  BC,0 ;SCRN TOP
15600         CALL POSIT
15700 PRNTX   LD  HL,TITLE
15800         LD  B,65 ;LENGTH
15900         CALL PRINT
16000         RET
```

```
16100 PAPRH CALL SLCT3 ;PRINTER
16200       JR  PRNTX
16300 ;************************
16400 ;   IN: HL-ADDR DATA
16500 ;       B -LENGTH DATA
16600 ;************************
16700 TV    PUSH HL
16800       PUSH BC
16900       CALL SLCT2 ;SCRN
17000       POP BC
17100       POP HL
17200       JR  PRINT
17300 ENDTV CALL SLCT2
17400       LD  A,13
17500       RST 16
17600       LD  A,(SPOSN+1)
17700       CP  2
17800       RET NZ
17900       LD  HL,#808E
18000       PUSH HL
18100       CALL #6200 ;SCROL
18200       CALL PRNTH
18300       LD  BC,#1500
18400       CALL POSIT
18500       RET
18600 ;************************
18700 ;  IN: HL-DATA ADDR
18800 ;      B -DATA LENGTH
18900 ;************************
19000 PRINT LD  A,(HL)
19100       RST 16
19200       INC HL
19300       DJNZ PRINT
19400       RET
19500 ;************************
19600 ;  IN: HL-DATA ADDR
19700 ;      B -DATA LENGTH
19800 ;************************
19900 PRNT1 PUSH HL
20000       PUSH BC
20100       LD A,1
20200       CALL SELCT
20300       POP BC
20400       POP HL
20500       CALL PRINT
20600       RET
20700 SLCT2 LD A,2
20800       JR SELCT
20900 SLCT3 LD A,3
21000       JR SELCT
21100 SELCT LD HL,#6029 ;SELECT
21200 PUSHH PUSH HL
21300       CALL #6200
21400       RET
21500 ;************************
21600 ;  IN: BC-LINE,COLUMN
21700 ;************************
21800 POSIT LD  HL,#801E ;SETAT
21900       JR  PUSHH
22000 ;************************
22100 ;  IN: HL-BINARY #
22200 ;  OUT: 5 DIGIT NUMBER TO
22300 ;       SCREEN
22400 ;************************
22500 NUMER RR  H
22600       RR  L
22700       PUSH AF
22800       LD  BC,5000
22900       CALL CNVRT
23000       LD  BC,500
23100       CALL CNVRT
23200       LD  BC,50
23300       CALL CNVRT
23400       LD  BC,5
23500       CALL CNVRT
23600       POP AF
23700       RL  L
23800       LD  BC,1
23900       CALL CNVRT
24000       RET
```

```
24100 CNVRT LD  A,0
24200 CNVT1 AND A
24300       SBC HL,BC
24400       JP  M,CNVT2
24500       INC A
24600       JR  CNVT1
24700 CNVT2 ADD HL,BC
24800       OR  #30
24900       RST 16
25000       RET
25100 WAIT  LD  B,100
25200 WAIT1 HALT
25300       DJNZ WAIT1
25400       RET
25500 STOPT LD  A,0
25600       JR  OUTTP
25700 STRT1 LD  A,1
25800       JR  OUTTP
25900 STRT2 LD  A,2
26000 OUTTP OUT (TADDR),A
26100       RET
26200 ERROR RST 8
26300 PTYPE DEFM /PROGRAM /
26400       DEFM /N.ARRAY /
26500       DEFM /C.ARRAY /
26600       DEFM /CODE    /
26700 TITLE DEFM /TYPE     NAME/
26800       DEFM /          ADDR /
26900       DEFM / SIZE   C/
27000 BLANK DEFM /            /
27100       DEFM /            /
27200       DEFM /          /
27300 EOFLN DEFB 13
27400 ASTER DEFM / */
27500 KCOPY DEFM /copy? /
27600 HSAVE DEFB 0
27700 SAVEL DEFW 0
27800 HTYPE DEFB 0
27900 HNAME DEFM /1234567890/
28000 HLEN  DEFW 0
28100 HADDR DEFW 0
28200       DEFW 0
28300 PROG  EQU 23635
28400 DFCC  EQU 23684
28500 SPOSN EQU 23688
28600 LASTK EQU 23560
28700 FLAGS EQU 23611
28800 TADDR EQU 127
```



'I think we have a mouse, too.'

```
5CD0 CD D7 5E 21 3D 80 E5 CD
5CD8 00 62 CD 35 5E CD 47 5E
5CE0 CD DB 5E CD 15 5E 37 DD
5CE8 21 51 5F 3E 00 11 11 00
5CF0 CD FC 00 30 F1 CD 27 5E
5CF8 3A 51 5F CB 27 CB 27 CB
5D00 27 4F 06 00 21 E5 5E ED
5D08 4A 06 08 CD 4C 5E 2A 84
5D10 5C 22 4F 5F 21 52 5F 06
5D18 0A CD 4C 5E 21 25 5F 06
5D20 01 CD 4C 5E CD 15 5E 37
5D28 DD 2A 53 5C 3E FF ED 5B
5D30 5C 5F CD FC 00 F5 CD 27
5D38 5E CD D7 5E F1 D2 AC 5D
5D40 3A 51 5F FE 00 20 11 3A
5D48 5F 5F CB 7F 28 0A 21 25
5D50 5F 06 06 CD 4C 5E 18 0E
5D58 2A 5E 5F CD 9A 5E 21 25
5D60 5F 06 01 CD 4C 5E 2A 5C
5D68 5F CD 9A 5E 21 48 5F 06
5D70 06 CD 78 5E 3E 8F 21 34
5D78 80 E5 CD 00 62 21 1A 80
5D80 E5 11 2C 01 21 C8 00 CD
5D88 00 62 76 3A 3B 5C CB 6F
5D90 28 EB CB AF 32 3B 5C 3A
5D98 08 5C F5 21 21 80 E5 CD
5DA0 00 62 F1 FE 59 28 0B FE
5DA8 79 CA B2 5D CD 55 5E C3
5DB0 E0 5C CD DF 5E CD D1 5E
5DB8 CD D1 5E CD 15 5E 37 DD
5DC0 21 51 5F 3E 00 11 11 00
5DC8 CD 68 00 D2 E4 5E CD 27
5DD0 5E CD D1 5E CD 15 5E 37
5DD8 DD 2A 53 5C 3E FF ED 5B
5DE0 5C 5F CD 68 00 D2 E4 5E
5DE8 CD 27 5E CD D7 5E 21 46
5DF0 5F 06 02 CD 4C 5E CD 55
5DF8 5E 06 08 2A 4F 5F 7D E6
5E00 E0 6F F3 E5 C5 11 24 80
5E08 D5 CD 00 62 C1 E1 24 10
5E10 F2 FB C3 E0 5C F3 DB FF
5E18 CB FF D3 FF DB F4 32 4E
5E20 5F 3E 01 D3 F4 FB C9 F3
```

```
5E28 3A 4E 5F D3 F4 DB FF CB
5E30 BF D3 FF FB C9 CD 85 5E
5E38 01 00 00 CD 95 5E 21 05
5E40 5F 06 41 CD 72 5E C9 CD
5E48 89 5E 18 F2 E5 C5 CD 85
5E50 5E C1 E1 18 1D CD 85 5E
5E58 3E 0D D7 3A 89 5C FE 02
5E60 C0 21 8E 80 E5 CD 00 62
5E68 CD 35 5E 01 00 15 CD 95
5E70 5E C9 7E D7 23 10 FB C9
5E78 E5 C5 3E 01 CD 8D 5E C1
5E80 E1 CD 72 5E C9 3E 02 18
5E88 04 3E 03 18 00 21 29 80
5E90 E5 CD 00 62 C9 21 1E 80
5E98 18 F6 CB 1C CB 1D F5 01
5EA0 88 13 CD C1 5E 01 F4 01
5EA8 CD C1 5E 01 32 00 CD C1
5EB0 5E 01 05 00 CD C1 5E F1
5EB8 CB 15 01 01 00 CD C1 5E
5EC0 C9 3E 00 A7 ED 42 FA CC
5EC8 5E 3C 18 F7 09 F6 30 D7
5ED0 C9 06 64 76 10 FD C9 3E
5ED8 00 18 06 3E 01 18 02 3E
5EE0 02 D3 7F C9 CF 50 52 4F
5EE8 47 52 41 4D 20 4E 2E 41
5EF0 52 52 41 59 20 43 2E 41
5EF8 52 52 41 59 20 43 4F 44
5F00 45 20 20 20 20 54 59 50
5F08 45 20 20 20 20 4E 41 4D
5F10 45 20 20 20 20 20 20 20
5F18 41 44 44 52 20 20 53 49
5F20 5A 45 20 20 43 20 20 20
5F28 20 20 20 20 20 20 20 20
5F30 20 20 20 20 20 20 20 20
5F38 20 20 20 20 20 20 20 20
5F40 20 20 20 20 20 0D 20 2A
5F48 63 6F 70 79 3F 20 00 00
5F50 00 00 31 32 33 34 35 36
5F58 37 38 39 30 00 00 00 00
5F60 00 00 00 00 00 00 00 00
5F68 00 00 00 00 00 00 00 00
5F70 00 00 00 00 00 00 00 00
5F78 00 00 00 00 00 00 00 00
```

TIMEX/SINCLIAR USERS GROUP
OF CINCINNATI
11 FUNSTON LN.
CINCINNATI, OHIO 45218